/*******************************************************************************
*   			Impacts of a Large-Scale Parenting Program: 				   *
*					Experimental Evidence from Chile						   *
********************************************************************************


	REQUIRES:	"${dt_final}panel_RR_indexes.dta"
	CREATES:	${out_tables}/T29A_heter_bycareopenness_big5.xls
				
	WRITEN BY:  Italo Lopez  [italolop@usc.edu]

********************************************************************************
	Prepare data
*******************************************************************************/

	* Load data
	use "${dt_final}/panel_RR_indexes.dta", clear


	* Establish macros for regressions

/*******************************************************************************
*   Set macros for regressions
*******************************************************************************/
	
	* Outcomes
	*-----------------------------------------------------------
	local childoutcomes  "zDCCSPunt zptevir_irtscore2 zt_score zcbcl_t_e zcbcl_t_i zibatt_sspi zibatt_ssai zibatt_sssr zas_raw_score" 
	local behaviors "home_score se_score disc_score" 
	local beliefs "zpstyle1 zpstyle2 zpstyle3 zPSCS zPSSS_family zPSSS_friends zPSSS_others zPSI_Distress zCESD zpacotis"	
	local child "zptevir_irtscore2 zcog_index zsoc_index" 

	* Controls
	*-----------------------------------------------------------

	macro  def Xvar1 "i.age_year i.gender"  	
	macro  def Xvar2 "i.age_year i.gender inc_qtaut_old i.pc_edu_mdsfin hh_mem hh_tipo"  	
	macro  def Xvar2old "i.gender inc_qtaut_old i.pc_edu_mdsfin_old hh_mem hh_tipo"  

	
	* Parental characteristics: WAIS and BFI and imputed wais 
	*-----------------------------------------------------------

	macro def parent_cha4 "BFI_ext BFI_ope BFI_con BFI_neu BFI_agr wais2 flag_wais"
	

/*******************************************************************************
Table A29: Heterogeneity of impact: Caregiver's Openness (Big Five)
*******************************************************************************/

forvalues num=2/3 {
    gen group`num'_lope=group`num'*lope
	gen group`num'_hope=group`num'*(1-lope)

	
	}



local finaloutcomes " zptevir_irtscore2 zcog_index zsoc_index" 
local list1 "finaloutcomes" 
#delimit ;
foreach f of local list1 {;

		foreach var of local `f' {;
					xi: xtreg `var' group2_lope group2_hope group3_lope group3_hope lope, fe i(CENTRO_SALUD) ;
				estimates store `var'_OPE;
				test group2_lope -group2_hope==0;
				estadd scalar p2=r(p);
				test group3_lope -group3_hope==0;
				estadd scalar p3=r(p);

				test group2_lope=group3_lope;
				estadd scalar lopep=r(p);
				test group2_hope =group3_hope;
				estadd scalar hopep=r(p);
				qui sum `var' if e(sample) & TIPO==1;
				estadd scalar mean_y=r(mean);	
				
				
				};
		
};

#delimit cr

  esttab zptevir_irtscore2_OPE zcog_index_OPE zsoc_index_OPE  using "${out_tables}/appendix/T29A_heter_bycareopenness_big5.csv", replace f ///	   
	    keep(group2_lope group2_hope group3_lope group3_hope) ///
		cells("b(fmt(3)star)" "se(fmt(3)par)") ///
	    star(* 0.10 ** 0.05 *** 0.01) ///
		stats(r2 p2 p3 N, fmt(2 3 3 0) ///
		labels("R-squared" "P-value for lope=hope: NEP-B" "P-value for lope=hope: NEP-I" "Observations")) legend ///
		varlabels(group2_lope "NEP-B Below median" group2_hope "NEP-B Above median" group3_lope "NEP-I  Below median" group3_hope "NEP-I Above median") ///
		mtitles("Vocabulary Index" "Exec. Function Index" "Socioemotional Index" )  								 
		
		
		
